
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 8.1.2.11 which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2013-03_D13k24’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_8_1_2_11
Test Group:
LTE_EPS_TS\8_1\RRC_EmergencyCall.ttcn
ATS Version:
iwd-EUTRA-B2013-03_D13wk24
System Simulator used:
R&S CMW500
UEs used:
Qualcomm MDM 8960 








and
Intel Mobile Communications XMM7160
Verification Status:
PASS


4. Corrections required for test case 8.1.2.11
This section describes the TTCN changes required to make EUTRA IMS Emergency Call test case 8.1.2.11 run correctly with an LTE UE. The TTCN used to pass this test case belongs to ‘iwd-EUTRA-B2013-03_D13wk24’ release.

Change 1 – Correction to function ‘f_TC_8_1_2_11_EUTRA’
	Function name
	f_TC_8_1_2_11_EUTRA()

	Reason for change
	Double configuration when calling the subsequent functions    ‘f_EUTRA_CellConfig_Def’ and ‘f_EUTRA_CellConfig_DRB_Common’.


	Summary of change
	Removed the first configuration (‘f_EUTRA_CellConfig_Def’) and modified the second by adding IMS Support setings to the broadcast SysInfos.


	TTCN module
	LTE_EPS_TS\8_1\RRC_EmergencyCall.ttcn

	MCC160 Comment
	


Before change

…
  function f_TC_8_1_2_11_EUTRA() runs on EUTRA_PTC

  {

    //@desc:     RRC Connention establishment of emergency call

    f_EUTRA_Init ( c1 );

    //Create and configure cell

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    // Create cell and send out system information

    f_EUTRA_CellConfig_DRB_Common(eutra_Cell1, 1, 1); // @sic R5-131470 sic@
    //Preamble to enter UE in E-UTRA RRC_IDLE (State 2)

    f_EUTRA_Preamble(eutra_Cell1,STATE2_IDLEUPDATE,-,2);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_2_11_Body();

    f_EUTRA_TestBody_Set(false);

    // Postamble

    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );

  } // End f_TC_8_1_2_11_EUTRA

…

After change

…
  function f_TC_8_1_2_11_EUTRA() runs on EUTRA_PTC

  {

    //@desc:     RRC Connention establishment of emergency call

    // Local variables delaration

    var Ims_EmergencySupport_r9 v_ImsEmergencySupported := true_;

    f_EUTRA_Init ( c1 );

    //Create and configure cell

    // f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    //Set Sys. info for Cell 1 according to 36.331

    f_EUTRA_CellInfo_SetSysInfo_EmergencySupport(eutra_Cell1, v_ImsEmergencySupported);
    // Create cell and send out system information

    f_EUTRA_CellConfig_DRB_Common(eutra_Cell1, 1, 1); // @sic R5-131470 sic@

    //Preamble to enter UE in E-UTRA RRC_IDLE (State 2)

    f_EUTRA_Preamble(eutra_Cell1,STATE2_IDLEUPDATE,-,2);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_2_11_Body();

    f_EUTRA_TestBody_Set(false);

    // Postamble

    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );

  } // End f_TC_8_1_2_11_EUTRA

…
Change 2 – Correction to local function ‘fl_TC_8_1_2_11_Body’
	Function name
	fl_TC_8_1_2_11_Body()

	Reason for change
	Incomplete synchronization between EUTRA-PTC and IMS-PTC.


	Summary of change
	Synchronization via new function ‘f_IMS_IPCAN_WaitForTriggerOnly’ (see also change 3).



	TTCN module
	LTE_EPS_TS\8_1\RRC_EmergencyCall.ttcn

	MCC160 Comment
	


Before change

…
  function fl_TC_8_1_2_11_Body()runs on EUTRA_PTC

  {

    //Steps 1 to 15 of the generic test procedure for IMS Emergency call establishment in EUTRA: Normal Service (TS 36.508 4.5A.4.3).

    //@siclog "Step 1-12" siclog@

    f_EUTRA_508_EstablishIMSEmergencyCall(eutra_Cell1, px_EmergencyCallNumber,"Step 2");

    //Wait Trigger from IMS

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN2]);
    //Release IMS Call

    f_UT_HangupCall(UT);

    //@siclog "Step 13" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.3 indicate that the UE is in E-UTRA RRC_CONNECTED state on Cell 1?

    f_EUTRA_508Check_ConnectedState ( eutra_Cell1 );

    }// End fl_TC_8_1_2_11_Body

…

After change

…
  function fl_TC_8_1_2_11_Body()runs on EUTRA_PTC

  {

    //Steps 1 to 15 of the generic test procedure for IMS Emergency call establishment in EUTRA: Normal Service (TS 36.508 4.5A.4.3).

    //@siclog "Step 1-12" siclog@

    f_EUTRA_508_EstablishIMSEmergencyCall(eutra_Cell1, px_EmergencyCallNumber,"Step 2");

    //Wait Trigger from IMS

    // f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN2]);

    f_IMS_IPCAN_WaitForTriggerOnly(IMS[tsc_Index_PDN2]); 
    //Release IMS Call

    f_UT_HangupCall(UT);

    f_IMS_IPCAN_WaitForTriggerOnly(IMS[tsc_Index_PDN2]); 

    //@siclog "Step 13" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.3 indicate that the UE is in E-UTRA RRC_CONNECTED state on Cell 1?

    f_EUTRA_508Check_ConnectedState ( eutra_Cell1 );

    }// End fl_TC_8_1_2_11_Body

…
Change 3 – Synchronization conflict resolved via new sync function
	Function name
	f_IMS_IPCAN_WaitForTriggerOnly()

	Reason for change
	Consider for example the test body in change 2 above: 
Once the emergency PDN is established the EUTRA-PTC shall wait for IMS signalling to complete and enter function WaitForTrigger before continuing. At the same time the IMS PTC code executes but in the middle of this, IMS PTC will ask the EUTRA PTC for the length of the authentication token by sending a special synchronization message to get this information. But this synchronization message was catched by function f_IMS_IPCAN_WaitForTrigger which fails and not by the default handler.


	Summary of change
	A new synchronization function was drafted which waits only for the specific sync template.



	TTCN module
	Common\IMS\IMS_PTC_CoordMsg.ttcn

	MCC160 Comment
	


Before change

After change

…
   function f_IMS_IPCAN_WaitForTriggerOnly(IMS_IPCAN_CO_ORD_PORT p_Port)

  {

      p_Port.receive(cmr_IMS_IPCAN_Trigger);

  }
…
Change 4 – Correction to function ‘f_TC_8_1_2_11_IMS2’
	Function name
	f_TC_8_1_2_11_IMS2()

	Reason for change
	- Synchronization messages are missing

- Callee url is not passed to function ‘f_IMS_CallReleaseMO’
- Deregistration is not required as per prose spec.


	Summary of change
	- Synchronization messages added

- Callee url  passed to function ‘f_IMS_CallReleaseMO’
- Deregistration removed.



	TTCN module
	LTE_EPS_TS\8_1\RRC_EmergencyCall_IMS.ttcn

	MCC160 Comment
	


Before change

…
  function f_TC_8_1_2_11_IMS2() runs on IMS_PTC

  {

    var REGISTER_Request v_RegisterRequest;

    var INVITE_Request v_InviteRequest;

    f_IMS_PTC_Init();

    // Test Case Body

    // Emergency registration: 34.229-1 Annex C.20

    v_RegisterRequest := f_IMS_EmergencyRegistration_C20();

    // Emergency call setup: 34.229-1 Annex C.22

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22();

    // Inform EUTRA: IMS signalling completed.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Release Emergency Call MO

    f_IMS_CallReleaseMO(v_InviteRequest);
    // Emergency DeRegistration

    f_IMS_PTC_Deregistration();
  }  
…

After change

…
  function f_TC_8_1_2_11_IMS2() runs on IMS_PTC

  {

    var REGISTER_Request v_RegisterRequest;

    var INVITE_Request v_InviteRequest;

    f_IMS_PTC_Init();

    // Test Case Body

    // Emergency registration: 34.229-1 Annex C.20

    v_RegisterRequest := f_IMS_EmergencyRegistration_C20();

    // Emergency call setup: 34.229-1 Annex C.22

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22();

    // Inform EUTRA: IMS signalling completed.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Release Emergency Call MO

    f_IMS_CallReleaseMO(v_InviteRequest,-,-, "urn:service:sos");  

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN); 

    // Emergency DeRegistration @olaf

    // f_IMS_PTC_Deregistration();

  }
…
Change 5 – Correction to function ‘f_IMS_InviteResponse_200_MessageHeaderTX’
	Function name
	f_IMS_InviteResponse_200_MessageHeaderTX()

	Reason for change
	The Contact header and P-AssertedId Headers shall refer to the callee Uri. In the case of emergency call the calleeUri is not the one stored in constant tsc_IM_CaleeContactUri, but the emergency uri.


	Summary of change
	The callee uri was passed as parameter by the calling function.

	TTCN module
	Common\IMS\IMS_CommonProcedures_CallControl.ttcn

	MCC160 Comment
	


Before change

…
  /*

   * @desc      200 OK for other requests than REGISTER or SUBSCRIBE (see TS 34.229, A.3.1) - INVITE

   * @param     p_InviteRequest

   * @param     p_IsGiba            (default value: false)

   * @param     p_ContentType       (default value: omit)

   * @param     p_IsEmergency       (default value: false)

   * @return    template (value) MessageHeader

   * @status    APPROVED_IMS

   */  

 function f_IMS_InviteResponse_200_MessageHeaderTX(INVITE_Request p_InviteRequest,

                                                    boolean p_IsGiba := false,

                                                    template (omit) ContentType p_ContentType := omit,

                                                    boolean p_IsEmergency := false) runs on IMS_PTC return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(p_InviteRequest.msgHeader, p_ContentType);

    v_MessageHeader_Response.recordRoute     := fl_IMS_RecordRoute_A23_A31_TX(p_IsGiba);

    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(tsc_IMS_CalleeContactUri));

    if (p_IsEmergency) {    // Condition A6

      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));

    }

    return v_MessageHeader_Response;

  }

…

After change

…
  /*

   * @desc      200 OK for other requests other than REGISTER or SUBSCRIBE (see TS 34.229, A.3.1) - INVITE

   * @param     p_InviteRequest

   * @param     p_IsGiba            (default value: false)

   * @param     p_ContentType       (default value: omit)

   * @param     p_IsEmergency       (default value: false)

   * @return    template (value) MessageHeader

   * @status    APPROVED_IMS

   */

  function f_IMS_InviteResponse_200_MessageHeaderTX(INVITE_Request p_InviteRequest,

                                                    boolean p_IsGiba := false,

                                                    template (omit) ContentType p_ContentType := omit,

                                                    boolean p_IsEmergency := false,

                                                    charstring p_CalleeContactUri := tsc_IMS_CalleeContactUri) runs on IMS_PTC return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(p_InviteRequest.msgHeader, p_ContentType);

    v_MessageHeader_Response.recordRoute     := fl_IMS_RecordRoute_A23_A31_TX(p_IsGiba);

    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_CalleeContactUri));

    if (p_IsEmergency) {    // Condition A6

      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(p_CalleeContactUri));

    }

    return v_MessageHeader_Response;

  }
…
Change 6 – Correction to function ‘f_IMS_AckRequest_MessageHeaderRX’
	Function name
	f_IMS_AckRequest_MessageHeaderRX()

	Reason for change
	34.229-1 Annex A.2.7 specifies for the To header 
“SIP URI of the SS when BYE is sent by the UE, but SIP URI of the UE when BYE is sent by the SS. URI must be the same as used for the endpoint in the earlier requests within the dialog.”. 
However, the To header is not correctly set here.


	Summary of change
	To header set to the earlier request To header if present.

	TTCN module
	Common\IMS\IMS_CommonProcedures_CallControl.ttcn

	MCC160 Comment
	


Before change

…
     function f_IMS_AckRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) StatusLine p_StatusLine_PrevResponse := omit,

                                            template (omit) MessageHeader p_MessageHeader_PrevResponse := omit) return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Ack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);

    var integer v_StatusCode;

    if (ispresent(p_StatusLine_PrevResponse)) {

      v_StatusCode := valueof(p_StatusLine_PrevResponse.statusCode);

      if ((v_StatusCode < 200) or (v_StatusCode > 299)) {                                     /* @sic R5-130682: ACK for non-2xx response sic@ */

        v_MessageHeader_Ack.via         := v_MessageHeader_Invite.via;

      } else {

        v_MessageHeader_Ack.via         := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); /* @sic R5-130682: ACK for 2xx response sic@ */

      }

    }

    v_MessageHeader_Ack.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_PrevResponse);

    v_MessageHeader_Ack.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!
    v_MessageHeader_Ack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Ack.cSeq            := cr_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Ack.pAccessNetworkInfo := omit;

    return v_MessageHeader_Ack;

  }

…

After change

…
  function f_IMS_AckRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) StatusLine p_StatusLine_PrevResponse := omit,

                                            template (omit) MessageHeader p_MessageHeader_EarlierRequest := omit) return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Ack := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_MessageHeader_GetCseqValue(v_MessageHeader_Invite);

    var integer v_StatusCode;

    if (ispresent(p_StatusLine_PrevResponse)) {

      v_StatusCode := valueof(p_StatusLine_PrevResponse.statusCode);

      if ((v_StatusCode < 200) or (v_StatusCode > 299)) {                                     /* @sic R5-130682: ACK for non-2xx response sic@ */

        v_MessageHeader_Ack.via         := v_MessageHeader_Invite.via;

      } else {

        v_MessageHeader_Ack.via         := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via); /* @sic R5-130682: ACK for 2xx response sic@ */

      }

    }

    if (ispresent(p_MessageHeader_EarlierRequest)) {
    v_MessageHeader_Ack.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_EarlierRequest);

    v_MessageHeader_Ack.toField         := p_MessageHeader_EarlierRequest.toField;  

   }
    v_MessageHeader_Ack.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Ack.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Ack.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Ack.cSeq            := cr_CseqDef(v_CSeqValue, "ACK");

    v_MessageHeader_Ack.maxForwards     := cr_MaxForwardsDef;

    v_MessageHeader_Ack.pAccessNetworkInfo := omit;

    return v_MessageHeader_Ack;

  }

…
Change 7 – Correction to function ‘f_IMS_ByeRequest_MessageHeaderRX’
	Function name
	f_IMS_ByeRequest_MessageHeaderRX()

	Reason for change
	34.229-1 Annex A.2.8 specify From and To headers 
“same value as received in the request”. 
However, From and To header is not correctly set.


	Summary of change
	Headers set with same value as in the previous request if present.

	TTCN module
	Common\IMS\IMS_CommonProcedures_CallControl.ttcn

	MCC160 Comment
	


Before change

…
  function f_IMS_ByeRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) MessageHeader p_MessageHeader_Prev183Response := omit,

                                            boolean p_IsGiba := false) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Bye := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMO);

    v_MessageHeader_Bye.via             := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via);

    v_MessageHeader_Bye.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_Prev183Response);

    v_MessageHeader_Bye.toField         := cr_ToWithTag(v_MessageHeader_Invite.toField.addressField, tsc_IMS_InviteToTag);     // !!!! CHECK !!!!
    v_MessageHeader_Bye.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Bye.cSeq            := cr_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cr_MaxForwardsDef;

    if (not p_IsGiba) {

      v_MessageHeader_Bye.require            := cr_Require;

      v_MessageHeader_Bye.proxyRequire       := cr_ProxyRequireDef;

      v_MessageHeader_Bye.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Bye.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    } else {

      v_MessageHeader_Bye.securityVerify     := omit;

    }

    return v_MessageHeader_Bye;

  }  
…

After change

…
    function f_IMS_ByeRequest_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) MessageHeader p_MessageHeader_PreviousRequest := omit,

                                            boolean p_IsGiba := false) runs on IMS_PTC return template (present) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Bye := cr_MessageHeader_Dummy;

    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogMO);

    v_MessageHeader_Bye.via             := fl_Via_AsInPrevInvite(v_MessageHeader_Invite.via);

    v_MessageHeader_Bye.route           := fl_Route_ReverseOrderOfRecordRoute(p_MessageHeader_Prev183Response);

     if (ispresent(p_MessageHeader_EarlierRequest)) {
    v_MessageHeader_Bye.toField         := p_MessageHeader_PreviousRequest.toField

    v_MessageHeader_Bye.fromField       := p_MessageHeader_PreviousRequest.fromField;

   }
    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.sessionId       := v_MessageHeader_Invite.sessionId;   /* @sic R5s130195 BASELINE MOVING 2013 - SessionId sic@ */

    v_MessageHeader_Bye.cSeq            := cr_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cr_MaxForwardsDef;

    if (not p_IsGiba) {

      v_MessageHeader_Bye.require            := cr_Require;

      v_MessageHeader_Bye.proxyRequire       := cr_ProxyRequireDef;

      v_MessageHeader_Bye.securityVerify     := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Bye.pAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    } else {

      v_MessageHeader_Bye.securityVerify     := omit;

    }

    return v_MessageHeader_Bye;

  }

…
Change 8 – Correction to function ‘f_IMS_CallReleaseMO’
	Function name
	f_IMS_CallReleaseMO()

	Reason for change
	Test case 8.1.2.11 triggers UT commands in the EUTRA PTC, whereas other test cases trigger UT commands in the IMS PTC or UTRAN PTC. A flag is needed to control the UT command. Also the callee uri needs to be sent in the request.


	Summary of change
	UT trigger flag and callee uri were passed as parameter.

	TTCN module
	Common\IMS\IMS_CommonProcedures_CallControl.ttcn

	MCC160 Comment
	


Before change

…
  function f_IMS_CallReleaseMO(INVITE_Request p_InviteRequest,

                               template (omit) MessageHeader p_MessageHeader_Prev183Response := omit,

                               boolean p_IsGIBA := false) runs on IMS_PTC

  { /* @sic R5s130333 change 6 - new parameter p_ContactUri sic@ */

    /* p_ContactUri: shall be present when p_MessageHeader_Prev183Response is omitted; overrules cantact of p_MessageHeader_Prev183Response if present */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var template (present) MessageHeader v_MessageHeader_Bye;

    var template (present) SipUrl v_ContactSipUrl;

    if (isvalue(p_MessageHeader_Prev183Response)) {

      v_ContactSipUrl := f_MessageHeader_GetContactSipUrl(valueof(p_MessageHeader_Prev183Response));

    } else {

      v_ContactSipUrl := f_SIP_BuildSipUri_RX(tsc_IMS_CalleeContactUri);

    }

    f_UT_IMS_CallRelease(MMI);   /* @sic R5s130333 change 20 sic@ */

    // --- Receive BYE

    v_MessageHeader_Bye := f_IMS_ByeRequest_MessageHeaderRX(p_InviteRequest, p_MessageHeader_Prev183Response, p_IsGIBA);

    IMS_Server.receive(car_IMS_Bye_Request(cr_BYE_Request(v_ContactSipUrl, v_MessageHeader_Bye))) -> value v_IMS_DATA_REQ;

    // --- Send 200 OK

    v_RoutingInfo := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo, cs_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_IMS_DATA_REQ.Request.Bye.msgHeader))));

}  
…

After change

…
  function f_IMS_CallReleaseMO(INVITE_Request p_InviteRequest,

                               template (omit) MessageHeader p_MessageHeader_Prev183Response := omit,

                               boolean p_IsGIBA := false,

                               charstring p_CalleeUri := tsc_IMS_CalleeContactUri,

                               boolean p_TriggerUT := false) runs on IMS_PTC

  { /* @sic R5s130333 change 6 - new parameter p_ContactUri sic@ */

    /* p_ContactUri: shall be present when p_MessageHeader_Prev183Response is omitted; overrules cantact of p_MessageHeader_Prev183Response if present */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var template (present) MessageHeader v_MessageHeader_Bye;

    var template (present) SipUrl v_ContactSipUrl;

    if (isvalue(p_MessageHeader_Prev183Response)) {

      v_ContactSipUrl := f_MessageHeader_GetContactSipUrl(valueof(p_MessageHeader_Prev183Response));

    } else {

      v_ContactSipUrl := f_SIP_BuildSipUri_RX(p_CalleeUri);

    }

    if (p_TriggerUT) {

    f_UT_IMS_CallRelease(MMI);   /* @sic R5s130333 change 20 sic@ */

    }
    // --- Receive BYE

    v_MessageHeader_Bye := f_IMS_ByeRequest_MessageHeaderRX(p_InviteRequest, p_MessageHeader_Prev183Response, p_IsGIBA);

    IMS_Server.receive(car_IMS_Bye_Request(cr_BYE_Request(v_ContactSipUrl, v_MessageHeader_Bye))) -> value v_IMS_DATA_REQ;

    // --- Send 200 OK

    v_RoutingInfo := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo, cs_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_IMS_DATA_REQ.Request.Bye.msgHeader))));

  }

…
Change 9 – Correction to function ‘f_IMS_ByeRequest_MessageHeaderRX’
	Function name
	f_IMS_ByeRequest_MessageHeaderRX()

	Reason for change
	The measurement parametrization in step 3 was not correct.

	Summary of change
	The measurement parametrization in step 3 is corrected. Note: this requires a prose CR.

	TTCN module
	Common\IMS\IMS_CommonProcedures_CallControl.ttcn

	MCC160 Comment
	


Before change

…
function f_IMS_EmergencyCallSetup_AnnexC22(boolean p_IsUnprotected := false,

                                             boolean p_GeolocationSupported := false) runs on IMS_PTC return INVITE_Request

  { /* !!!! acc. to A.2.1 in step 1 MIME encapsulation is used only acc. to condition A8 !!!! */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template(value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var charstring v_CalleeUri := "urn:service:sos";                                   /* Request-Uri 34.229-1 Annex A.2.1 (FFS)

                                                                                          !!!! acc. to RFC 5031 there are sub-services defined; it is not clear whether or in which case the UE will use such a sub-service !!!! */

    var template (present) SipUrl v_CalleeSipUri := f_SIP_BuildSipUri_RX(v_CalleeUri);

    var INVITE_Request v_InviteRequest;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) SDP_Message v_SDP_MessageStep1;

    var template (value) SDP_Message v_SDP_MessageStep4;

    var SDP_Message v_SDP_Message_Step1;

    var template (present) MessageBody v_MessageBody_Step1;

    var boolean v_IsEmergency := true;

    var boolean v_PrackRequiredFor180Ringing := false;  /* => there is no PRACK expected to acknowledge the 180 response */

    var charstring v_FmtAudio;

    var charstring v_XmlBody;

    var bitstring v_BitStr;

    var Geolocation_Type v_Geolocation;

    v_SDP_MessageStep1 := f_IMS_BuildSDP_AnnexC22_Step1();

    if (p_GeolocationSupported) {

      v_MessageBody_Step1 := cr_MessageBody_MIME(cr_MIME_Message_EmergencyCall(v_SDP_MessageStep1));

    } else {

      v_MessageBody_Step1 := cr_MessageBody_SDP(v_SDP_MessageStep1);

    }

    // Step 1. Receive INVITE for an Emergency call

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(-, p_IsUnprotected);

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest(cr_INVITE_Request(v_CalleeSipUri, v_MessageHeader_Invite, v_MessageBody_Step1), v_IsEmergency, p_GeolocationSupported);

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    if (p_GeolocationSupported) {

      if (lengthof(v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList) != 2){

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Invite body");

      }

      if (ischosen (v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[0].mime_encapsulated_part.sdpMessageBody)){

        v_SDP_Message_Step1 := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[0].mime_encapsulated_part.sdpMessageBody;

        v_XmlBody := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[1].mime_encapsulated_part.xmlBody;

      } else {

        v_SDP_Message_Step1 := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[1].mime_encapsulated_part.sdpMessageBody;

        v_XmlBody := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[0].mime_encapsulated_part.xmlBody;

      }

      v_BitStr := oct2bit(char2oct(v_XmlBody));

      if (decvalue(v_BitStr, v_Geolocation) != 0) {

        FatalError(__FILE__, __LINE__, "Location info can not be decoded.");

      }

      if (not match (v_Geolocation, fl_IMS_GeolocationInfoRx())) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid geolocation info");

      }

    } else {

      v_SDP_Message_Step1 := v_InviteRequest.messageBody.sdpMessageBody;

    }

    v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(v_SDP_Message_Step1.media_list[0], omit);   // !!!! PROSE ISSUE: prose does not specify how to get the correct fmt !!!!

    if (v_FmtAudio == "") {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "no AMR codec found in given media description");

    }

    // Step 2. SS sends a 100 Trying provisional response.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // Step 3. SS sends a 180 Ringing.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, -, -, v_PrackRequiredFor180Ringing, v_IsEmergency))));

    // Step 4. SS responds to the invite INVITE with 200 OK including the SDP record.

    v_SDP_MessageStep4 := f_IMS_BuildSDP_AnnexC22_Step4(v_FmtAudio, v_SDP_Message_Step1.origin.addr);

    v_MessageHeader_InviteResponse_200 := f_IMS_InviteResponse_200_MessageHeaderTX(v_InviteRequest, -, cs_ContentTypeDef("application/sdp"));
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200, cs_MessageBody_SDP(v_SDP_MessageStep4))));

    // Step 5. Finally, UE acknowledges call setup.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, v_MessageHeader_InviteResponse_200);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeSipUri, v_MessageHeader_Ack)));

    return v_InviteRequest;

  }

…

After change

…
function f_IMS_EmergencyCallSetup_AnnexC22(boolean p_IsUnprotected := false,

                                             boolean p_GeolocationSupported := false) runs on IMS_PTC return INVITE_Request

  { /* !!!! acc. to A.2.1 in step 1 MIME encapsulation is used only acc. to condition A8 !!!! */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template(value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var charstring v_CalleeUri := "urn:service:sos";                                   /* Request-Uri 34.229-1 Annex A.2.1 (FFS)

                                                                                          !!!! acc. to RFC 5031 there are sub-services defined; it is not clear whether or in which case the UE will use such a sub-service !!!! */

    var template (present) SipUrl v_CalleeSipUri := f_SIP_BuildSipUri_RX(v_CalleeUri);

    var INVITE_Request v_InviteRequest;

    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (present) MessageHeader v_MessageHeader_Invite;

    var template (present) SDP_Message v_SDP_MessageStep1;

    var template (value) SDP_Message v_SDP_MessageStep4;

    var SDP_Message v_SDP_Message_Step1;

    var template (present) MessageBody v_MessageBody_Step1;

    var boolean v_IsEmergency := true;

    var boolean v_PrackRequiredFor180Ringing := false;  /* => there is no PRACK expected to acknowledge the 180 response */

    var charstring v_FmtAudio;

    var charstring v_XmlBody;

    var bitstring v_BitStr;

    var Geolocation_Type v_Geolocation;

    v_SDP_MessageStep1 := f_IMS_BuildSDP_AnnexC22_Step1();

    if (p_GeolocationSupported) {

      v_MessageBody_Step1 := cr_MessageBody_MIME(cr_MIME_Message_EmergencyCall(v_SDP_MessageStep1));

    } else {

      v_MessageBody_Step1 := cr_MessageBody_SDP(v_SDP_MessageStep1);

    }

    // Step 1. Receive INVITE for an Emergency call

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(-, p_IsUnprotected);

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest(cr_INVITE_Request(v_CalleeSipUri, v_MessageHeader_Invite, v_MessageBody_Step1), v_IsEmergency, p_GeolocationSupported);

    v_InviteRequest := v_IMS_DATA_REQ.Request.Invite;

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo);

    if (p_GeolocationSupported) {

      if (lengthof(v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList) != 2){

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Invite body");

      }

      if (ischosen (v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[0].mime_encapsulated_part.sdpMessageBody)){

        v_SDP_Message_Step1 := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[0].mime_encapsulated_part.sdpMessageBody;

        v_XmlBody := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[1].mime_encapsulated_part.xmlBody;

      } else {

        v_SDP_Message_Step1 := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[1].mime_encapsulated_part.sdpMessageBody;

        v_XmlBody := v_InviteRequest.messageBody.mimeMessageBody.mimeEncapsulatedList[0].mime_encapsulated_part.xmlBody;

      }

      v_BitStr := oct2bit(char2oct(v_XmlBody));

      if (decvalue(v_BitStr, v_Geolocation) != 0) {

        FatalError(__FILE__, __LINE__, "Location info can not be decoded.");

      }

      if (not match (v_Geolocation, fl_IMS_GeolocationInfoRx())) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid geolocation info");

      }

    } else {

      v_SDP_Message_Step1 := v_InviteRequest.messageBody.sdpMessageBody;

    }

    v_FmtAudio := f_SDP_MediaDescr_GetAMR_8000_1(v_SDP_Message_Step1.media_list[0], omit);   // !!!! PROSE ISSUE: prose does not specify how to get the correct fmt !!!!

    if (v_FmtAudio == "") {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "no AMR codec found in given media description");

    }

    // Step 2. SS sends a 100 Trying provisional response.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine100, f_IMS_InviteResponse_100_MessageHeaderTX(v_InviteRequest))));

    // Step 3. SS sends a 180 Ringing.

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, -, -, v_PrackRequiredFor180Ringing, v_IsEmergency))));

    // Step 4. SS responds to the invite INVITE with 200 OK including the SDP record.

    v_SDP_MessageStep4 := f_IMS_BuildSDP_AnnexC22_Step4(v_FmtAudio, v_SDP_Message_Step1.origin.addr);

    v_MessageHeader_InviteResponse_200 := f_IMS_InviteResponse_200_MessageHeaderTX(v_InviteRequest, -, 

                    cs_ContentTypeDef("application/sdp"), false, v_CalleeUri);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200, cs_MessageBody_SDP(v_SDP_MessageStep4))));

    // Step 5. Finally, UE acknowledges call setup.

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, v_MessageHeader_InviteResponse_200);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_CalleeSipUri, v_MessageHeader_Ack)));

    return v_InviteRequest;

  }

…
Change 10 – Correction to local function in ‘fl_IMS_ContactAddress_CheckParams’
	Function name
	fl_IMS_ContactAddress_CheckParams()

	Reason for change
	The SOS parameters check in the case of emergency calls is checking an incorrect parameter list.


	Summary of change
	Checked the Url parameter list instead of the contact parameter list for SOS parameters compliance.



	TTCN module
	Common\IMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…
    function fl_IMS_ContactAddress_CheckParams(ContactAddress p_ContactAddress,

                                             boolean p_CondA7 := false,

                                             boolean p_IMS_MTSI := tsc_IMS_MTSI,

                                             boolean p_IMS_GRUUsInSIP := pc_IMS_GRUUsInSIP,

                                             boolean p_IMS_SMSoverIMS := tsc_IMS_SMSoverIMS,

                                             boolean p_Video := tsc_IMS_VideoFeature) runs on IMS_PTC

  { /* NOTE: 'Expires' can only be checked in context with the 'Expires' header field */

    var SemicolonParam_List v_ContactParams := {};

    if (ispresent(p_ContactAddress.contactParams)) {

      v_ContactParams := p_ContactAddress.contactParams;

    }

    if (p_CondA7) {

      if (not match(v_ContactParams, cr_ReqTypeParamInContactParams("sos"))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }
    // Rel-7 specific, @sic R5-073440 sic@

    if (p_IMS_MTSI) {                                                             // check 'feature-param'

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue))) { // @sic R5s130109 sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'Contact Feature Params'");

      }

    }

    if (p_IMS_GRUUsInSIP) {                                                       // check 'c-p-instance'

      if (not match(v_ContactParams, cr_CPInstanceInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_SMSoverIMS) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_SMSParamInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid SMS params in 'ContactAddress'");

      }

    }

    if (p_Video) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid video params in 'ContactAddress'");

      }

    }

  }
…

After change

…
  function fl_IMS_ContactAddress_CheckParams(ContactAddress p_ContactAddress,

                                             boolean p_CondA7 := false,

                                             boolean p_IMS_MTSI := tsc_IMS_MTSI,

                                             boolean p_IMS_GRUUsInSIP := pc_IMS_GRUUsInSIP,

                                             boolean p_IMS_SMSoverIMS := tsc_IMS_SMSoverIMS,

                                             boolean p_Video := tsc_IMS_VideoFeature) runs on IMS_PTC

  { /* NOTE: 'Expires' can only be checked in context with the 'Expires' header field */

    var SemicolonParam_List v_ContactParams := {};

    var SemicolonParam_List v_UrlParams := {};
    if (ispresent(p_ContactAddress.contactParams)) {

      v_ContactParams := p_ContactAddress.contactParams;

    }

    if (p_CondA7) {

      if (ispresent(p_ContactAddress.addressField.nameAddr.addrSpec.urlParameters)) {

         v_UrlParams := p_ContactAddress.addressField.nameAddr.addrSpec.urlParameters;

      }

      if (not match(v_UrlParams, cr_SemicolonParam_List_OneSpecificParam("sos"))) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid sos Url params in 'ContactAddress'");

      }

    }
    // Rel-7 specific, @sic R5-073440 sic@

    if (p_IMS_MTSI) {                                                             // check 'feature-param'

      if (not match(v_ContactParams, cr_FeatureParamInContactParams(tsc_IMS_AcceptContactValue))) { // @sic R5s130109 sic@

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'Contact Feature Params'");

      }

    }

    if (p_IMS_GRUUsInSIP) {                                                       // check 'c-p-instance'

      if (not match(v_ContactParams, cr_CPInstanceInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid params in 'ContactAddress'");

      }

    }

    if (p_IMS_SMSoverIMS) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_SMSParamInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid SMS params in 'ContactAddress'");

      }

    }

    if (p_Video) {                                                       // check 'feature-param'

      if (not match(v_ContactParams, cr_VideoInContactParams)) {

        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid video params in 'ContactAddress'");

      }

    }

  }

…
Change 11 – Correction to local function ‘fl_SIP_ParseURI’
	Function name
	fl_SIP_ParseURI()

	Reason for change
	The regular expression for parsing the sos uri is incorrect.


	Summary of change
	Regular expression corrected: String separator is ‘:’ instead of ‘@’.

	TTCN module
	Common\IMS\IMS_SIP_Templates.ttcn

	MCC160 Comment
	


Before change

…
    function fl_SIP_ParseURI(charstring p_URI) return template (value) SipUrl

  { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

    var template (present) charstring v_Pattern;

    var template (value) UriComponents v_UriComponents;

    var template (omit) UserInfo v_UserInfo := omit;

    var charstring v_Host;

    var charstring v_Scheme;

    var charstring v_HierPart;

    var charstring v_UserInfoString;

    var charstring v_UserName;

    v_Pattern  := pattern "([a-z]#(3,4)):(*)";                  // @sic R5s130187 change 8: "pattern" sic@

    v_Scheme   := regexp(p_URI, v_Pattern, 0);

    v_HierPart := regexp(p_URI, v_Pattern, 1);

    select (v_Scheme) {

      case (c_sipScheme) {                                      // sip uri acc. to RFC 3261 cl. 19.1.1

        v_Pattern  := pattern "(*)@(*)";                        // @sic R5s130187 change 8: "pattern" sic@

        v_UserInfoString := regexp(v_HierPart, v_Pattern, 0);

        if (v_UserInfoString == "") {                           // sip uri does not have user info e.g. "sip:biloxi.com" or sip:[2001:db8:1::139]

          v_Host := v_HierPart;

        } else {                                                // sip uri has user info

          v_Host := regexp(v_HierPart, v_Pattern, 1);

          v_Pattern  := pattern "(*):(*)";                      // @sic R5s130187 change 8: "pattern" sic@

          v_UserName := regexp(v_UserInfoString, v_Pattern, 0);

          if (v_UserName == "") {                               // user info but no password e.g. "sip:alice@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserInfoString);

          } else {                                              // user info with password e.g. "sip:alice:secretword@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserName, regexp(v_UserInfoString, v_Pattern, 1));

          }

        }

        v_UriComponents := cs_SipUriComponent(v_UserInfo, v_Host);

      }

      case (c_telScheme) {                                      // tel uri acc. to RFC 3966

        v_UriComponents := cs_TelUriComponent(v_HierPart);

      }

      case (c_urnScheme) {                                      // urn uri for emergency calls e.g. "urn:service:sos"

        v_Pattern  := pattern "(*)@(*)";
        v_UriComponents := cs_UrnUriComponent(regexp(v_HierPart, v_Pattern, 0), regexp(v_HierPart, v_Pattern, 1));

      }

      case else {

        v_UriComponents := cs_OtherComponent(v_HierPart);

      }

    }

    return cs_SipUrl_Common(v_Scheme, v_UriComponents);

  }

…

After change

…
  function fl_SIP_ParseURI(charstring p_URI) return template (value) SipUrl

  { /* @sic R5s130195 BASELINE MOVING 2013: change of SipUrl sic@ */

    var template (present) charstring v_Pattern;

    var template (value) UriComponents v_UriComponents;

    var template (omit) UserInfo v_UserInfo := omit;

    var charstring v_Host;

    var charstring v_Scheme;

    var charstring v_HierPart;

    var charstring v_UserInfoString;

    var charstring v_UserName;

    v_Pattern  := pattern "([a-z]#(3,4)):(*)";                  // @sic R5s130187 change 8: "pattern" sic@

    v_Scheme   := regexp(p_URI, v_Pattern, 0);

    v_HierPart := regexp(p_URI, v_Pattern, 1);

    select (v_Scheme) {

      case (c_sipScheme) {                                      // sip uri acc. to RFC 3261 cl. 19.1.1

        v_Pattern  := pattern "(*)@(*)";                        // @sic R5s130187 change 8: "pattern" sic@

        v_UserInfoString := regexp(v_HierPart, v_Pattern, 0);

        if (v_UserInfoString == "") {                           // sip uri does not have user info e.g. "sip:biloxi.com" or sip:[2001:db8:1::139]

          v_Host := v_HierPart;

        } else {                                                // sip uri has user info

          v_Host := regexp(v_HierPart, v_Pattern, 1);

          v_Pattern  := pattern "(*):(*)";                      // @sic R5s130187 change 8: "pattern" sic@

          v_UserName := regexp(v_UserInfoString, v_Pattern, 0);

          if (v_UserName == "") {                               // user info but no password e.g. "sip:alice@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserInfoString);

          } else {                                              // user info with password e.g. "sip:alice:secretword@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserName, regexp(v_UserInfoString, v_Pattern, 1));

          }

        }

        v_UriComponents := cs_SipUriComponent(v_UserInfo, v_Host);

      }

      case (c_telScheme) {                                      // tel uri acc. to RFC 3966

        v_UriComponents := cs_TelUriComponent(v_HierPart);

      }

      case (c_urnScheme) {                                      // urn uri for emergency calls e.g. "urn:service:sos"

        v_Pattern  := pattern "(*):(*)";
        v_UriComponents := cs_UrnUriComponent(regexp(v_HierPart, v_Pattern, 0), regexp(v_HierPart, v_Pattern, 1));

      }

      case else {

        v_UriComponents := cs_OtherComponent(v_HierPart);

      }

    }

    return cs_SipUrl_Common(v_Scheme, v_UriComponents);

  }

…
Change 12 – Correction to function ‘f_IMS_BuildSDP_AnnexC22_Step1’
	Function name
	f_IMS_BuildSDP_AnnexC22_Step1()

	Reason for change
	The expected template for the SDP record in Annex C.22 Step 1 is too stringent expecting particular fmtp contents. This check is not according to 34.229-1 Annex C.22 step 1 which does not specify fmtp contents. 

Also the check of bandwidth is incorrect: The UE will send bandwidth information not only for AS but also for RS and RR streams.


	Summary of change
	A new template for receiving emergency SDP records was defined,  

‘cr_SDP_Emc_Media_Audio’, see change 13.



	TTCN module
	Common\IMS\IMS_SDP_Messages.ttcn

	MCC160 Comment
	


Before change

…
  function f_IMS_BuildSDP_AnnexC22_Step1() runs on IMS_PTC return template (present) SDP_Message

  { /* Build SDP record to be received from the UE at step 1 according to 34.229-1 Annex C.22 */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template SDP_attribute_list v_SDP_Attribute_List := *;

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Media_Audio(-, ?, ?),
                                          cr_SDP_Connection(v_AddrType, v_IPAddrOrigin),

                                          cr_SDP_Bandwidth_List_AS(?),
                                          v_SDP_Attribute_List);

    return cr_SDP_Message_AnySession(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }

…

After change

…
  function f_IMS_BuildSDP_AnnexC22_Step1() runs on IMS_PTC return template (present) SDP_Message

  { /* Build SDP record to be received from the UE at step 1 according to 34.229-1 Annex C.22 */

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var template (present) charstring v_IPAddrOrigin := ?;

    var template (present) charstring v_IPAddrConnection := ?;

    var template (present) SDP_media_desc v_SDP_Media_Desc;

    var template SDP_attribute_list v_SDP_Attribute_List := *;

    v_SDP_Media_Desc := cr_SDP_Media_Desc(cr_SDP_Emc_Media_Audio,
                                          cr_SDP_Connection(v_AddrType, v_IPAddrOrigin),

                                          cr_SDP_Bandwidth_List_Media(?,?,?),
                                          v_SDP_Attribute_List);

    return cr_SDP_Message_AnySession(v_AddrType, v_IPAddrOrigin, v_IPAddrConnection, v_SDP_Media_Desc);

  }

…
Change 13 – Definition of new template ‘cr_SDP_Emc_Media_Audio’
	Template name
	cr_SDP_Emc_Media_Audio

	Reason for change
	Template for the SDP record in the Invite message  Annex C.22 Step 1 is missing.


	Summary of change
	New template defined.

	TTCN module
	Common\IMS\IMS_SDP_Templates.ttcn

	MCC160 Comment
	


Before change

After change

…
   // Emergency call template for media field , see Annex C.22

   template (present) SDP_media_field cr_SDP_Emc_Media_Audio :=

  /* @status    APPROVED_IMS */

    cr_SDP_Media(c_audio, ?, ?, cr_SDP_FmtList_AtLeastOne(?));
…
Change 14 – Correction to function ‘f_EUTRA_Get_508_TrafficFlowTemplate’
	Function name
	f_EUTRA_Get_508_TrafficFlowTemplate()

	Reason for change
	PDN index is not passed to the function. For emergency calls we need to consider that we are on the second PDN with different IP addresses.


	Summary of change
	PDN index was passed to the main functions and secondary functions. Also comments and references to 36.508 section 6.6.2 were corrected.



	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

…
  function f_EUTRA_Get_508_TrafficFlowTemplate(integer p_BearerContext,

                                               HalfOctet_Type p_EPS_BearerId := tsc_EpsDedicatedBearerId,

                                               template (omit) B3_Type p_PdnTypeValue := omit) runs on EUTRA_PTC return template (value) TrafficFlowTemplate

  { /* @sic R5-112600: to replace cs_508_Tft_Dedicated_1 and cs_508_Tft_Dedicated_2 sic@ */

    var integer v_PacketFilterLen1ByRef;

    var integer v_PacketFilterLen2ByRef;

    var template (value) PacketFilter v_PacketFilter1ByRef;

    var template (value) PacketFilter v_PacketFilter2ByRef;

    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;

    var integer v_IeLength;

    var B3_Type v_PdnTypeValue;

    if (isvalue(p_PdnTypeValue)) {

      v_PdnTypeValue := valueof(p_PdnTypeValue);

    } else {

      v_PdnTypeValue := f_EUTRA_MobileInfo_GetAssignedPdnType();

    }

    select (p_BearerContext) {

      case (1) {        // @sic R5-123632 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 1, p_EPS_BearerId, v_PdnTypeValue);  // get packet filter #1 acc. to 36.508 Table 6.6.2-2

        f_Get_508_PacketFilter(v_PacketFilter2ByRef, v_PacketFilterLen2ByRef, 2, p_EPS_BearerId, v_PdnTypeValue);  // get packet filter #2 acc. to 36.508 Table 6.6.2-3

        v_IeLength := v_PacketFilterLen1ByRef + v_PacketFilterLen2ByRef + 1;  // packet filter 1 + packet filter 2 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef, v_PacketFilter2ByRef};

      }

      case (2) {        // @sic R5-123632, R5-132085 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 3, p_EPS_BearerId, v_PdnTypeValue);  // get packet filter #3 acc. to 36.508 Table 6.6.2-4

        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef};

      }

      case (3) {        // @sic R5-132085 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 5, p_EPS_BearerId, v_PdnTypeValue);  // get packet filter #3 acc. to 36.508 Table 6.6.2-4

        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef};

      }

      case (4) {        // @sic R5-132085 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 4, p_EPS_BearerId, v_PdnTypeValue);  // get packet filter #3 acc. to 36.508 Table 6.6.2-4

        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef};

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid bearer context");

      }

    }

    return cs_TrafficFlowTemplate(v_IeLength, v_PacketFilterList);

  }

…

After change

…
  function f_EUTRA_Get_508_TrafficFlowTemplate(integer p_BearerContext,

                                               HalfOctet_Type p_EPS_BearerId := tsc_EpsDedicatedBearerId,

                                               template (omit) B3_Type p_PdnTypeValue := omit,

                                               PDN_Index_Type p_PdnIndex := PDN_1) runs on EUTRA_PTC return template (value) TrafficFlowTemplate

  { /* @sic R5-112600: to replace cs_508_Tft_Dedicated_1 and cs_508_Tft_Dedicated_2 sic@ */

    var integer v_PacketFilterLen1ByRef;

    var integer v_PacketFilterLen2ByRef;

    var template (value) PacketFilter v_PacketFilter1ByRef;

    var template (value) PacketFilter v_PacketFilter2ByRef;

    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;

    var integer v_IeLength;

    var B3_Type v_PdnTypeValue;

    if (isvalue(p_PdnTypeValue)) {

      v_PdnTypeValue := valueof(p_PdnTypeValue);

    } else {

      v_PdnTypeValue := f_EUTRA_MobileInfo_GetAssignedPdnType();

    }

    select (p_BearerContext) {

      case (1) {        // @sic R5-123632 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 1, p_EPS_BearerId, v_PdnTypeValue, p_PdnIndex);  // get packet filter #1 acc. to 36.508 Table 6.6.2-2

        f_Get_508_PacketFilter(v_PacketFilter2ByRef, v_PacketFilterLen2ByRef, 2, p_EPS_BearerId, v_PdnTypeValue, p_PdnIndex);  // get packet filter #2 acc. to 36.508 Table 6.6.2-3

        v_IeLength := v_PacketFilterLen1ByRef + v_PacketFilterLen2ByRef + 1;  // packet filter 1 + packet filter 2 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef, v_PacketFilter2ByRef};

      }

      case (2) {        // @sic R5-123632, R5-132085 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 3, p_EPS_BearerId, v_PdnTypeValue, p_PdnIndex);  // get packet filter #2 acc. to 36.508 Table 6.6.2-2
        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef};

      }

      case (3) {        // @sic R5-132085 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 5, p_EPS_BearerId, v_PdnTypeValue, p_PdnIndex);  // get packet filter #3 acc. to 36.508 Table 6.6.2-4
        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef};

      }

      case (4) {        // @sic R5-132085 sic@

        f_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 4, p_EPS_BearerId, v_PdnTypeValue, p_PdnIndex);  // get packet filter #4 acc. to 36.508 Table 6.6.2-5
        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter

        v_PacketFilterList := {v_PacketFilter1ByRef};

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid bearer context");

      }

    }

    return cs_TrafficFlowTemplate(v_IeLength, v_PacketFilterList);

  }
…
Change 15 – Correction to function ‘f_EUTRA_508_EstablishIMSEmergencyCall’
	Function name
	f_EUTRA_508_EstablishIMSEmergencyCall()

	Reason for change
	1.    For emergency calls we have following EPS bearer IDs: Default EPS on first PDN is 5, default EPS on second PDN is 6, and dedicated EPS for speech data is 7. The currently present TTCN settings are incorrect.


2. 
A flag is needed for handling UT commands at the EUTRA-PTC level. Test case 8.1.2.11 triggers UT commands in the EUTRA-PTC, other test cases trigger UT commands in the IMS PTC or even in the UTRAN PTC.

3.    FFS comments still present.



	Summary of change
	1.    EPS bearer IDs were corrected and handled in main function to enable a more generic management of IDs for future emergency test cases with ongoing speech and/or video calls. 


2.    Flag for UT commands added.

3. 
FFS comments removed.


	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

…
  function f_EUTRA_508_EstablishIMSEmergencyCall(EUTRA_CellId_Type p_CellId,

                                                 charstring p_NumToDial := px_EmergencyCallNumber,

                                                 charstring p_StepNo := "") runs on EUTRA_PTC

  {

    var NAS_MSG_Indication_Type v_NasInd;

    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;

    // Step 1

    f_UT_RequestIMSCall (UT, p_NumToDial);

    // Steps 2 - 4

    v_NasInd := f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_4(p_CellId,

                                                           //    v_EpsBearerId2, // FFS
                                                               p_StepNo);

    // Steps 5 - 8

    f_EUTRA_508_EstablishIMSEmergencyCall_Steps5_8(p_CellId,

                                                //   v_EpsBearerId2, // FFS
                                                   v_NasInd);

    // Step 9

    v_NasInd := f_EUTRA_508_EstablishIMSEmergencyCall_Step9(p_CellId); // FFS
    // Steps 10 - 12

    f_EUTRA_508_EstablishIMSEmergencyCall_Steps10_12(p_CellId,

                                                     v_EpsBearerId2, // FFS
                                                     v_NasInd);

    // Steps 13 - 14

    f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14(p_CellId);

    // Step 15

    f_EUTRA_508_EstablishIMSEmergencyCall_Step15(p_CellId);

  } // f_EUTRA_508_EstablishIMSEmergencyCall

…

After change

…
  function f_EUTRA_508_EstablishIMSEmergencyCall(EUTRA_CellId_Type p_CellId,

                                                 charstring p_NumToDial := px_EmergencyCallNumber,

                                                 charstring p_StepNo := "", boolean p_TriggerUT := true ) runs on EUTRA_PTC

  {

    var NAS_MSG_Indication_Type v_NasInd;

    var EPS_BearerIdentity v_EpsDefBearerId2 := '6'H;

    var EPS_BearerIdentity v_EpsDedicatedBearerId2 := '7'H;
    // Step 1

    if (p_TriggerUT) {
    f_UT_RequestIMSCall (UT, p_NumToDial);

    }
    // Steps 2 - 4

    v_NasInd := f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_4(p_CellId,

                                                               p_StepNo);

    // Steps 5 - 8

    f_EUTRA_508_EstablishIMSEmergencyCall_Steps5_8(p_CellId,

                                                   v_NasInd);

    // Step 9

    v_NasInd := f_EUTRA_508_EstablishIMSEmergencyCall_Step9(p_CellId); 

    // Steps 10 - 12

    f_EUTRA_508_EstablishIMSEmergencyCall_Steps10_12(p_CellId,

                                                     v_EpsDefBearerId2, 

                                                     v_NasInd);

    // Steps 13 - 14

    f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14(p_CellId,

                                                     v_EpsDefBearerId2,

                                                     v_EpsDedicatedBearerId2);
    // Step 15

    f_EUTRA_508_EstablishIMSEmergencyCall_Step15(p_CellId,

                                                 v_EpsDedicatedBearerId2);

  } // f_EUTRA_508_EstablishIMSEmergencyCall 

…
Change 16 – Correction to function ‘f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14’
	Function name
	f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14()

	Reason for change
	1.   Default and dedicted bearer IDs need to be passed as parameters, these parameters are managed by higher functions.


2.   Table 4.5A.4.3-1: EUTRA/EPS signalling for IMS Emergency Call step 13 refers to EPS context #4. Incorrect context is used.


3.   PDN index is not passed to the functions generating the appropriate packet filters.


	Summary of change
	1.   Default and dedicted bearer IDs passed to and used by the current function. 

2.   Correct EPS context #4 used.

3.   PDN-2 passed to function _EUTRA_Get_508_TrafficFlowTemplate.


	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

…
    function f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    var EPS_BearerIdentity v_EpsDedicatedBearerId := '7'H;
    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReq :=

      cs_508_ActDedicatedEPSBearerCxtReq(v_EpsDedicatedBearerId,

                                         tsc_PTI_Unassigned,

                                         cs_LinkedId(hex2bit( tsc_EpsDefaultBearerId)),

                                         cs_508_EPS_QoS_Dedicated_1,

                                         f_EUTRA_Get_508_TrafficFlowTemplate(1, v_EpsDedicatedBearerId));
    var template (value) NAS_MSG_RequestList_Type v_NAS_MsgList := {

      cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_ActDedicatedEPSBearerCxtReq)

    };

    // Steps 13-14

    f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig(p_CellId,

                                                                 tsc_RRC_TI_Def,

                                                                 {tsc_DRB3},

                                                                 v_NAS_MsgList,

                                                                 UM_Mode);

  } // f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14

…

After change

…
    function f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14(EUTRA_CellId_Type p_CellId,

                                                            EPS_BearerIdentity p_EpsDefBearerId, 

                                                            EPS_BearerIdentity p_EpsDedicatedBearerId, 

                                                            ) runs on EUTRA_PTC

  {

    /*  Use Context #4

    *   See Table 4.5A.4.3-1: EUTRA/EPS signalling for IMS Emergency Call step 13

    */
    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReq :=

      cs_508_ActDedicatedEPSBearerCxtReq(p_EpsDedicatedBearerId,
                                         tsc_PTI_Unassigned,

                                         cs_LinkedId(hex2bit( p_EpsDefBearerId),

                                         cs_508_EPS_QoS_Dedicated_1,

                                         f_EUTRA_Get_508_TrafficFlowTemplate(4, p_EpsDedicatedBearerId), PDN_2); // Use context #4, PDN_2

    var template (value) NAS_MSG_RequestList_Type v_NAS_MsgList := {

      cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_ActDedicatedEPSBearerCxtReq)

    };

    // Steps 13-14

    f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig(p_CellId,

                                                                 tsc_RRC_TI_Def,

                                                                 {tsc_DRB3},

                                                                 v_NAS_MsgList,

                                                                 UM_Mode);

  } // f_EUTRA_508_EstablishIMSEmergencyCall_Steps13_14
…
Change 17 – Correction to function ‘f_EUTRA_508_EstablishIMSEmergencyCall_Step15’
	Function name
	f_EUTRA_508_EstablishIMSEmergencyCall_Step15()

	Reason for change
	Dedicated EPS bearer ID is hardcoded.


	Summary of change
	Dedicated EPS bearer ID passed as a parameter instead.

	TTCN module
	Common\EUTRA\EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

…
function f_EUTRA_508_EstablishIMSEmergencyCall_Step15(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC

  {

    var EPS_BearerIdentity v_EpsDedicatedBearerId := '7'H;
    // Step 15

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(v_EpsDedicatedBearerId))));

  } // f_EUTRA_508_EstablishIMSEmergencyCall_Step15
…

After change

…
  function f_EUTRA_508_EstablishIMSEmergencyCall_Step15(EUTRA_CellId_Type p_CellId

                                               EPS_BearerIdentity p_EpsDedicatedBearerId, 

) runs on EUTRA_PTC

  {

    // Step 15

    SRB.receive(car_SRB2_NasPdu_IND(p_CellId,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActDedicatedEPSBearerCxtAccept(p_EpsDedicatedBearerId))));

  } // f_EUTRA_508_EstablishIMSEmergencyCall_Step15

…
Change 18 – Correction to function ‘f_Get_508_PacketFilter’
	Function name
	f_Get_508_PacketFilter()

	Reason for change
	As stated in its comment, this function shall deliver the packet filters according to 36.508 Table 6.6.2-2 for the different EPS bearer contexts. But the coding of these packet filters was not according to the lastest 36.508 reference. 


	Summary of change
	Packet filters corrected according to TS 36.508 Table 6.6.2-2 to 6.6.2-6, taking into account additional corrections. The latter may require a Prose CR to 36.508.

Media ports used and linked to the values used in the IMS SDP records.


	TTCN module
	Common\NAS\ NAS_AuxiliaryDefsAndFunctions.ttcn

	MCC160 Comment
	


Before change

…
   //----------------------------------------------------------------------------

  /*

   * @desc      Get the Dedicated TFT according to 36.508 clause 6.6.2

   * @param     p_PacketFilterByRef (by reference)

   * @param     p_PacketFilterLengthByRef (by reference)

   * @param     p_PacketFilterNumber ... is the reference packet filter in Table 6.6.2-2
   * @param     p_EPS_BearerId

   * @param     p_PdnTypeValue

   * @param     p_PdnIndex   (default: PDN_1)

   * @status    APPROVED

   */

 function f_Get_508_PacketFilter(out template (value) PacketFilter p_PacketFilterByRef,

                                  out integer p_PacketFilterLengthByRef,

                                  integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId,

                                  B3_Type p_PdnTypeValue,

                                  PDN_Index_Type p_PdnIndex := PDN_1)

  { /* @sic R5-112600 new parameter p_PdnTypeValue to determine whether IPv4 or IPv6 address to be used sic@ */

    /* @sic R5s130362 - MCC160 Implementation: tsc_PDN_AddressInfo; new parameter p_PdnIndex sic@ */

    var integer v_ContentsLength := 0;

    var template (value) PacketFilter v_Filter;

    var integer v_Index := enum2int(p_PdnIndex);

    var PDN_AddressInfo_Type v_PDN_AddressInfo := tsc_PDN_AddressInfo[v_Index];

    var charstring v_NetworkIpAddrv4 := v_PDN_AddressInfo.PCSCF_IPAddressIPv4;

    var charstring v_NetworkIpAddrv6 := v_PDN_AddressInfo.PCSCF_IPAddressIPv6;

    if ((p_PacketFilterNumber != 4) or (p_PacketFilterNumber != 5)) {   // @sic R5-132085 sic@ in case (4, 5) no addresses used
      if (p_PdnTypeValue == tsc_PdnType_IPv4) {  // IPv4 only; @sic R5s110168 change 8.3; R5-112600 sic@

        v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress(v_NetworkIpAddrv4);

        v_ContentsLength := v_ContentsLength + 9;

      }

      else { // IPv4v6 or IPv6 only

        v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress(v_NetworkIpAddrv6);

        v_ContentsLength := v_ContentsLength + 33;

      }

    }
    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.direction   := '01'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.iei         := '0'H;  //  @sic R5s130195 Baseline Moving sic@

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);

        v_ContentsLength := v_ContentsLength + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_ContentsLength := v_ContentsLength + 2;

      }

      case (2) {

        // Update to values in Reference packet filter #2

        // UL only filter, ID=1          @sic R5-104702 sic@

        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.direction   := '10'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.iei         := '1'H;  //  @sic R5s130195 Baseline Moving sic@

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6+8), 8); // @sic R5-104702 sic@

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);

        v_ContentsLength := v_ContentsLength + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_ContentsLength := v_ContentsLength + 2;

      }

      case (3){

        // Update to values in Reference packet filter #3

        // Bidirectional filter, ID=2  @sic R5-104702 sic@

        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.direction   := '11'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.iei         := '2'H;  //  @sic R5s130195 Baseline Moving sic@

        v_Filter.precedence  := '00001111'B;

      }

      case (4){  // @sic R5-132085 sic@

        // Update to values in Reference packet filter #4

        // Bidirectional filter, ID=3

        v_Filter.spare       := '00'B;

        v_Filter.direction   := '11'B;

        v_Filter.iei         := '3'H;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(60350, 60450);  // used same numbers as in 10.9.1

        v_ContentsLength := v_ContentsLength + 5;

        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeader;
        v_ContentsLength := v_ContentsLength + 2;

      }

      case (5){  // @sic R5-132085 sic@

        // Update to values in Reference packet filter #5

        // Bidirectional filter, ID=3

        v_Filter.spare       := '00'B;

        v_Filter.direction   := '11'B;

        v_Filter.iei         := '4'H;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(60350, 60450);  // used same numbers as in 10.9.1

        v_ContentsLength := v_ContentsLength + 5;

        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeader;
        v_ContentsLength := v_ContentsLength + 2;

      }

   }

    v_Filter.iel := int2oct(v_ContentsLength, 1);

    p_PacketFilterByRef := v_Filter;

    p_PacketFilterLengthByRef := v_ContentsLength + 3; // length of the filter contents + 3 octets for iei, precedence and iel

  }

…

After change

…
  //----------------------------------------------------------------------------

  /*

   * @desc      Get the Dedicated TFT according to 36.508 clause 6.6.2

   * @param     p_PacketFilterByRef (by reference)

   * @param     p_PacketFilterLengthByRef (by reference)

   * @param     p_PacketFilterNumber ... is the reference packet filter in Tables 6.6.2-2 to 6.6.2-6
   * @param     p_EPS_BearerId

   * @param     p_PdnTypeValue

   * @param     p_PdnIndex   (default: PDN_1)

   * @status    APPROVED

   */

  function f_Get_508_PacketFilter(out template (value) PacketFilter p_PacketFilterByRef,

                                  out integer p_PacketFilterLengthByRef,

                                  integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId,

                                  B3_Type p_PdnTypeValue,

                                  PDN_Index_Type p_PdnIndex := PDN_1)

  { /* @sic R5-112600 new parameter p_PdnTypeValue to determine whether IPv4 or IPv6 address to be used sic@ */

    /* @sic R5s130362 - MCC160 Implementation: tsc_PDN_AddressInfo; new parameter p_PdnIndex sic@ */

    var integer v_ContentsLength := 0;

    var template (value) PacketFilter v_Filter;

    var integer v_Index := enum2int(p_PdnIndex);

    var PDN_AddressInfo_Type v_PDN_AddressInfo := tsc_PDN_AddressInfo[v_Index];

    var charstring v_NetworkIpAddrv4 := v_PDN_AddressInfo.PCSCF_IPAddressIPv4;

    var charstring v_NetworkIpAddrv6 := v_PDN_AddressInfo.PCSCF_IPAddressIPv6;

    var integer v_Speech_MediaPortNumber := tsc_IMS_MediaPort_M1;

    var integer v_Video_MediaPortNumber := tsc_IMS_MediaPort_M2;
if ((p_PacketFilterNumber == 1) or (p_PacketFilterNumber == 2) or (p_PacketFilterNumber == 3)) {   // @sic R5-132085 sic@ in case (4, 5) no addresses used

      if (p_PdnTypeValue == tsc_PdnType_IPv4) {  // IPv4 only; @sic R5s110168 change 8.3; R5-112600 sic@

        v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress(v_NetworkIpAddrv4);

        v_ContentsLength := v_ContentsLength + 9;

      }

      else { // IPv4v6 or IPv6 only

        v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress(v_NetworkIpAddrv6);

        v_ContentsLength := v_ContentsLength + 33;

      }

    }

    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.direction   := '01'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.iei         := '0'H;  //  @sic R5s130195 Baseline Moving sic@

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);

        v_ContentsLength := v_ContentsLength + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_ContentsLength := v_ContentsLength + 2;

      }

      case (2) {

        // Update to values in Reference packet filter #2

        // UL only filter, ID=1          @sic R5-104702 sic@

        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.direction   := '10'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.iei         := '1'H;  //  @sic R5s130195 Baseline Moving sic@

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6+8), 8); // @sic R5-104702 sic@

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);

        v_ContentsLength := v_ContentsLength + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_ContentsLength := v_ContentsLength + 2;

      }

      case (3){

        // Update to values in Reference packet filter #3

        // Bidirectional filter, ID=2  @sic R5-104702 sic@

        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.direction   := '11'B; //  @sic R5s130195 Baseline Moving sic@

        v_Filter.iei         := '2'H;  //  @sic R5s130195 Baseline Moving sic@

        v_Filter.precedence  := '00001111'B;

      }

      case (4){  // @sic R5-132085 sic@

        // Update to values in Reference packet filter #4

        // Bidirectional filter, ID=3

        v_Filter.spare       := '00'B;

        v_Filter.direction   := '11'B;

        v_Filter.iei         := '3'H;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(v_Speech_MediaPortNumber, v_Speech_MediaPortNumber+1); 

        v_ContentsLength := v_ContentsLength + 5;

        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeader;
        v_ContentsLength := v_ContentsLength + 2;

      }

      case (5){  // @sic R5-132085 sic@

        // Update to values in Reference packet filter #5

        // Bidirectional filter, ID=3

        v_Filter.spare       := '00'B;

        v_Filter.direction   := '11'B;

        v_Filter.iei         := '4'H;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(v_Video_MediaPortNumber, v_Video_MediaPortNumber+1);  

        v_ContentsLength := v_ContentsLength + 5;

        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeader;
        v_ContentsLength := v_ContentsLength + 2;

      }

   }

    v_Filter.iel := int2oct(v_ContentsLength, 1);

    p_PacketFilterByRef := v_Filter;

    p_PacketFilterLengthByRef := v_ContentsLength + 3; // length of the filter contents + 3 octets for iei, precedence and iel

  }

…
Change 19 – Correction to IMS constants
	Constant name
	tsc_IMS_MediaPort_M1  ;
tsc_IMS_MediaPort_M2  ;
tsc_IMS_MediaPort_M3  ;

	Reason for change
	According to the RTP RFC 3550 section 11: 
“RTP relies on the underlying protocol(s) to provide demultiplexing of RTP data and RTCP control streams.  For UDP and similar protocols,RTP SHOULD use an even destination port number and the corresponding RTCP stream SHOULD use the next higher (odd) destination port number.”


However, RTP media port numbers are not even.


	Summary of change
	RTP media ports set as even integers.

	TTCN module
	Common\IMS\IMS_CommonDefs.ttcn

	MCC160 Comment
	


Before change

…
  const integer tsc_IMS_MediaPort_M1 := 50061;
  const integer tsc_IMS_MediaPort_M2 := 50063;
  const integer tsc_IMS_MediaPort_M3 := 50065;

…

After change

…
  const integer tsc_IMS_MediaPort_M1 := 60350; // RTP port Speech

  const integer tsc_IMS_MediaPort_M2 := 60352; // RTP port Video

  const integer tsc_IMS_MediaPort_M3 := 60354; // RTP port Text
…
Change 20 – Correction to ‘f_SDP_MediaDescr_GetAMR_8000_1’
	Function name
	f_SDP_MediaDescr_GetAMR_8000_1()

	Reason for change
	As stated in its comment, the function shall deliver the AMR codec for the media description in the SDP payload. According to 34.229 Annex C.22 Step 1, AMR codec shall be present, but no specific value is expected. 


	Summary of change
	AMR codec corrected according to RFC 4867 section 8, channel number is optional and may not be sent, in this case default value has to be considered i.e 1.



	TTCN module
	Common\IMS\IMS_SDP_Templates.ttcn

	MCC160 Comment
	


Before change

…

function f_SDP_MediaDescr_GetAMR_8000_1(SDP_media_desc p_SDP_Media_Desc,

                                          template (omit) CharStringList_Type p_FmtpParameters := tsc_AMR_Fmtp_DefaultParameters) return charstring

  {

    return f_SDP_MediaDescr_GetMediaFormat(p_SDP_Media_Desc, tsc_CODEC_AMR_8000_1, p_FmtpParameters);

  }

…
After change

…

function f_SDP_MediaDescr_GetAMR_8000_1(SDP_media_desc p_SDP_Media_Desc,

                                          template (omit) CharStringList_Type p_FmtpParameters := tsc_AMR_Fmtp_DefaultParameters) return charstring

  {

    return f_SDP_MediaDescr_GetMediaFormat(p_SDP_Media_Desc, (tsc_CODEC_AMR_8000_1, "AMR/8000"), p_FmtpParameters);

  }

…
Change 21 – Correction to template ‘cr_SDP_Message_AnySession’
	Template name
	cr_SDP_Message_AnySession()

	Reason for change
	The template is a check for the validity of SDP payload in the receive direction.
The template is not updated according to latest version of 34.229-1. According to 34.229-1 Annex C.22 Step1 (INVITE), bandwidth info in session description of the SDP payload is optional, but in the current TTCN template it needs to be present.


	Summary of change
	Bandwidth info in session description of SDP payload set to “Any or Omit”.

	TTCN module
	Common\IMS\IMS_SDP_Templates.ttcn

	MCC160 Comment
	


Before change

… 

template (present) SDP_Message cr_SDP_Message_AnySession(template (present) charstring p_AddrType := ?,

                                                           template (present) charstring p_IPAddrOrigin := ?,

                                                           template (present) charstring p_IPAddrConnection := ?,

                                                           template (present) SDP_media_desc p_SDP_Media_Desc) :=

  /* @status    APPROVED_IMS */

    cr_SDP_Message_Common(p_AddrType, p_IPAddrOrigin, p_IPAddrConnection, -, -, cr_SDP_Time_Any, { p_SDP_Media_Desc });

}
…
After change

…

template (present) SDP_Message cr_SDP_Message_AnySession(template (present) charstring p_AddrType := ?,

                                                           template (present) charstring p_IPAddrOrigin := ?,

                                                           template (present) charstring p_IPAddrConnection := ?,

                                                           template (present) SDP_media_desc p_SDP_Media_Desc) :=

  /* @status    APPROVED_IMS */

    cr_SDP_Message_Common(p_AddrType, p_IPAddrOrigin, p_IPAddrConnection, -, *, cr_SDP_Time_Any, { p_SDP_Media_Desc });

}
…
5. Execution Log Files

5.1 Qualcomm MDM 8960 UE

The Qualcomm MDM 8960 UE passed this test case on R&S CMW500 in LTE FDD band 13. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
Rohde-Schwarz\tc_8_1_2_11_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5.2 Intel Mobile Communications XMM7160 UE 

The Intel Mobile Communications XMM7160 UE passed this test case on R&S CMW500 in LTE FDD band 20. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
Rohde-Schwarz\tc_8_1_2_11_Intel.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	[1]
	R5s130496:    Supporting information for agreement of TC 8.1.2.11. This archive comprises:

                        -  html and text format execution log file

                        (Please note the embedded draft Prose CR)

